PharmaSUG China 2024 – Paper 10027

Return to baseline status of the worst elevated liver function tests

Yuan Sun, Tingting Zhang, Sanofi

ABSTRACT

The serum concentration of TBL (total bilirubin), ALP (alkaline phosphatase), ALT (alanine aminotransferase) and AST (aspartate aminotransferase) are often used for estimating drugs' liver injury in safety analysis. The most common ways are comparing incidences and the worst post baseline results of increased serum concentration from different arms. If the worst post baseline result is several times the upper limit of normal reference range, it may be a signal of liver cells injury.

However, the temporary increased serum concentration about TBL, ALP, ALT or AST is not a reliable predictor of a drug's potential for liver injury. If the serum concentration temporarily raises but then drops to the baseline level, it may be considered that the drug will not cause liver damage and provide evidence of drug safety.

This paper is to track the status of the worst elevated liver function tests in a dynamic view. Based on whether the post baseline serum concentration of TBL, ALP, ALT, and AST return to the baseline status or not, we can estimate drugs' liver injury.

INTRODUCTION

The human liver is the main organ for drug metabolism and detoxification. Over the past 50 years, DILI (Drug-Induced Liver Injury) has been the primary reason for drug safety withdrawal, thus liver function analysis has been an important part of drug safety analysis. Generally speaking, liver damage that leads to severe DILI is mainly caused by liver cell damage. Damaged liver cells release Alanine or Aspartate aminotransferase (ALT or AST), which in turn causes an increase in the concentration of ALT and AST in the serum. Therefore, the serum concentration of ALT and ALS is often used to evaluate the effect of drugs on the liver. In addition, other indicators, such as serum total bilirubin (TBL) and alkaline phosphatase (ALP) are also used to measure drug toxicity to the liver. Increased serum concentrations of these indicators during treatment mean that the drug may cause liver toxicity. When their serum concentrations are several times the normal value, the higher the multiple, the greater the possibility that the drug will cause liver toxicity.

In the current safety analysis, the main method about liver function is comparing incidences and the worst post baseline results of increased serum concentration for indicators like ALT, AST, ALT, TBL…. from different arms. Taking ALT as an example, count the number of subjects whose ALT serum concentrations is 5 times higher than the normal concentration from different groups during treatment , and calculate the incidence for every group. Drug groups with higher incidence rates may have greater effects on the liver.

However, an increase in serum concentration of indicators like ALT, AST, ALP, TBL cannot fully explain that the drug is hepatotoxic. Firstly, many studies have shown that a variety of drugs can cause a temporary increase in the serum concentrations of the above indicators. Secondly, changes in the serum concentrations are also related to the health status of the subjects, especially the subjects' own liver function. If a subject's ALT serum concentration is significantly higher than normal level without treatment, it is not sufficient to indicate that the drug is hepatotoxic, even if the ALT serum concentration increases multiple times during treatment. Therefore, only observing the temporary change about serum concentration of ALT, AST, ALP, TBL cannot accurately explain the effect of the drug on the liver. If we can consider both the changes in serum concentrations of ALT, AST, ALP, TBL over the whole study period in a dynamic view, and the status of the subject's liver function, we can more comprehensively evaluate the safety of the drug.

RETURN TO BASELINE STATUS OF THE WORST ELEVATED LIVER FUNCTION TESTS

In safety analysis, the comparation between baseline status and post baseline status is often used to evaluate drugs' potential safety issues. If the serum concentration temporarily raises but then drops to the baseline level, it may be considered that the drug will not cause liver damage and provide evidence of drug safety.

Because subjects may experience many times of serum concentration elevation about ALT, AST, ALP, TBL. The maximum value means the worst influence on the liver. Therefore, it is most conservative to observe and evaluate the changes in the maximum concentration of the indicators during treatment.

This paper mainly describes the dynamic observation of whether the maximum concentrations of ALT, AST, ALP, and TBL during treatment return to their baseline levels as treatment progresses and provide evidence of drug safety.

FIVE SITUATIONS OF RETURN TO BASELINE STATUS

As treatment progresses, the worst outcome (indicators' maximum serum concentration) returns to the baseline status can be divided into five situations as shown in Table 1. In the first and second column, variables NORMEDN and NORMED are used to stand for return to baseline status. The third column provides the author's simple interpretation of these five situations. The minus sign (-) means that it cannot be evaluated, and the more plus signs (+), the greater the impact of the drug on the liver.

Table 1 Five situations in which ALT, AST, ALP, and TBL return to baseline status

NORMEDN	NORMED	Safety of drugs
0	No worsening as compared to the baseline status	-
1	Returned to baseline status before last IMP dose	+
2	Returned to baseline status after last IMP dose	++
3	Never returned to baseline status	+++
4	No assessment after elevation	-

DETAILED EXPLANATION OF THE FIVE RETURN SITUATIONS

In order to describe the five situations of return to the baseline status more intuitively, in this section, we take ALT as an example, using graphs and corresponding dummy data to explain the details. These dummy data are also the source data for running the SAS code in the next section.

In order to describe the five situations of return to the baseline status more intuitively, in this section, we take ALT as an example, using graphs and corresponding dummy data to explain the details. These dummy data are also the source data for running the SAS code in the next section.

SYMBOLS IN FIGURE

Firstly, we need to understand the meaning of letters and numbers in figures. ULN stands for the upper limit of normal range of ALT serum concentration. Assume that when x is 0, 1, 2, 3 and 4, it represents ALT concentrations within the normal range, >3ULN, >5ULN, >10ULN and >20ULN, respectively. Bx stands for the baseline status of ALT serum concentration. For example, B0 stands for subject's ALT serum concentration is normal, while B1 stands for the ALT serum concentration is abnormal: it is greater than the value of 3 times ULN. Similarly, Tx represents the ALT concentration during the medication period (on treatment period), and Px represents the ALT level after the medication period (post treatment period). Last OT value is the last ALT serum concentration in the on treatment period, which we assume as the time range of the last medication plus one day. TE stands for the treatment emergent. Last max TE value is the last max ALT serum concentration in the treatment emergent period. In general , on treatment period represents the time of taking drugs, usually defined as the last dose date plus 1 day. But treatment emergent period covers the different time in different studies. It may be the last dose date plus one day, or several months. Therefore, the Last max TE value may appear during the on treatment period, or in the

post treatment period.

VARIABLES IN DUMMY DATA

Next, we will interpret the variables in dummy data. ANRHI represents the upper limit of the normal range of ALT. AVAL is the specific concentration of ALT in each test. ADT and ADM are the time of each test. MCRIT3MN and MCRIT1MN represent the baseline ALT status and the post baseline ALT status, respectively. ABLFL='Y' represents that this record is a baseline record. LMTEFL refers to the Last Max TE value mentioned above. LVOTFL represents the Last OT value. ONTRTFL represents On treatment period. For the specific methods of these flag variables, please refer to Table 2.

Table 2 Variables that need to be defined in advance

VARIABLE	LABEL	RULE
ONTRTFL	On treatment period	The period between the first IMP to the last IMP+1 day.
TRTEMFL	Treatment emergent period	The period between the first IMP to the last IMP + n days. (n is specified by studies)
LMTEFL	Last max treatment emergent value	• Sort all the post baseline ALT status in treatment emergent period by: 1. ALT status value (AVAL), 2. ALT test day (ADT and ATM). • Find the last max value and give flag to the value.
LVOTFL	Last on treatment value	• Sort all the ALT status (AVAL) in on treatment period by ALT test day. • Find the last value of ATL test and give flag to the value.
MCRIT3MN	Baseline ALT status	• Assign 0 if ALT <= 3ULN (AVAL<=3*ANRHI); • Assign 1 if ALT > 3ULN; • Assign 2 if ALT > 5ULN; • Assign 3 if ALT > 10ULN; • Assign 4 if ALT > 20ULN;
MCRIT1MN	Post baseline ALT status	

EXPLANATION

For better understanding, we assume that in all the following scenarios, the baseline status is ALT>3ULN, and the treatment emergent period covers the overall period of both on treatment period and post treatment period.

1. No worsening as compared to the baseline status

In Figure 1 A, the last max ALT concentration is also >3ULN. But as the progress of taking drugs, the ALT concentration decreases, and becomes to normal status before the last time of taking drugs. Observing the entire period, we can see that the ALT level of the subject is not worse than the baseline status.

Figure 1. No worsening as compared to the baseline status

The data of subject 1 in Table 3 simulates the situation in Figure 1 A.

Table 3 Dummy data for figure 1

SUBJID	ANRHI	AVAL	MCRIT3 MN	MCRIT1 MN	ADT	ATM	ABLFL	ONTRT FL	LMTE FL	LVOT FL
1	50	60	1	1	2023-01-01	10:00	Y			
1	50	60	1	1	2023-01-06	11:00		Y	Y	
1	50	40	1	0	2023-01-11	12:00		Y		
1	50	40	1	0	2023-01-17	13:00		Y		Y

2. Returned to baseline status before last IMP dose

As shown in Figure 2 A, although the subject's last max ALT concentration increases to >5ULN in on treatment period, which is worse than the baseline status. But after that, the ALT concentration decreases and drops to >3ULN level in the last IMP dose cycle, which means the ALT concentration returns to baseline status before last IMP dose.



Figure 2. Returned to baseline status before last IMP dose

The data of subject 2 in Table 4 simulates the situation in Figure 2 A.

Table 4 Dummy data for figure 2

SUBJID	ANRHI	AVAL	MCRIT3 MN	MCRIT1 MN	ADT	ATM	ABLFL	ONTRT FL	LMTE FL	LVOT FL
2	50	60	1	1	2024-01-01	10:00	Y			
2	50	110	1	2	2024-01-06	11:00		Y	Y	
2	50	60	1	1	2024-01-11	12:00		Y		Y
2	50	60	1	1	2024-01-17	13:00				

3. Returned to baseline status after last IMP dose

Compared with Figure 2 A, in figure 3 A, the last max ALT concentration (>5ULN), appeared in the last IMP dose cycle. Then in post treatment period, it drops to >3ULN. We consider it as return to baseline status after last IMP dose.

In figure 3 B, observing the on treatment period, we could see a signal of return to baseline before last IMP: from B1 to T1. But the last max TE value P3 (>10ULN) is actually in the post treatment period. So, checking from the whole treatment emergent period, compared to baseline status, the ALT concentration firstly becomes worse (P3), then drops to normal (P0) in post treatment period, thus it is also a scenario of return to baseline status after last IMP dose.

Figure 3. Returned to baseline status after last IMP dose

The data of subject 3 in Table 5 simulates the situation in Figure 3 B.

Table 5 Dummy data for figure 5

SUBJID	ANRHI	AVAL	MCRIT3 MN	MCRIT1 MN	ADT	ATM	ABLFL	ONTRT FL	LMTE FL	LVOT FL
3	50	60	1	1	2024-01-22	14:00	Y			
3	50	40	1	0	2024-01-27	15:00		Y		
3	50	60	1	1	2024-02-01	16:00		Y		Y
3	50	160	1	3	2024-02-06	17:00			Y	
3	50	60	1	0	2024-02-11	18:00				

4. Never returned to baseline status

In Figure 4 A, the subject's ALT concentration increases to >5ULN in on treatment period. During the post treatment period, ALT concentration increases again to >10ULN and then drops back to >5ULN. Throughout the process, the ALT concentration remains higher than the baseline status (>3ULN). This situation can be easily detected as never returned to the baseline status.

In figure 4 B, it has a similar trend as figure 3 B, but the ALT concentration (P2) after the last max TE value is still higher than the baseline status, Thus it is also a scenario of never returned to baseline status.



Figure 4. Never returned to baseline status

The data of subject 4 in Table 6 simulates the situation in Figure 4 A.

Table 6 Dummy data for figure 4

SUBJID	ANRHI	AVAL	MCRIT3 MN	MCRIT1 MN	ADT	ATM	ABLFL	ONTRT FL	LMT EFL	LVOT FL
4	50	60	1	1	2023-01-22	14:00	Y			
4	50	110	1	2	2023-01-27	15:00		Y		
4	50	120	1	2	2023-02-01	16:00		Y		Y
4	50	170	1	3	2023-02-06	17:00			Y	

SUBJID	ANRHI	AVAL	MCRIT3 MN	MCRIT1 MN	ADT	ATM	ABLFL	ONTRT FL	LMT EFL	LVOT FL
4	50	110	1	2	2023-02-11	18:00				

5. No assessment after elevation

In the examples shown in Figure 5 A and 5 B, the subjects don't have further tests after the last max ALT concentration values. Therefore, we cannot determine whether the ALT concentration returns to the baseline status or not.



Figure 5. No assessment after elevation

The data of subject 5 in Table 7 simulates the situation in Figure 5 B.

Table 7 Dummy data for figure 5

SUBJID	ANRHI	AVAL	MCRIT3 MN	MCRIT1 MN	ADT	ATM	ABLFL	ONTRT FL	LMTE FL	LVOT FL
5	50	60	1	1	2023-01-11	12:00	Y			
5	50	110	1	2	2023-01-17	13:00		Y	Y	Y

SAS CODING

In this section we will display the SAS code for generating variables NORMED and NORMEDN about return to baseline status. The flag variables which are needed in code like LMTEFL, ONTRTFL can be created in advance per rules in table 2.

Step 1: Import source data. Dummy data in table 3 to table 7 are the source data. You can copy them in one table in EXCEL and import the EXCEL file in SAS.

```
* #1 import data;
proc import datafile="xxx/NORMTEST.xlsx"
        out=test
        dbms=xlsx
        replace;
        getnames=yes;
run;
```

Step 2: Sort data by subject ID, parameter code, and tests date of indicators.

```
* #2 sort data by subject ID, parameter, and test start date;
proc sort data=test out=test1;
   by subjid paramcd adt atm;
run;
```

Step 3: Determine the ALT level of each record one by one, from the last max TE value (records of LMTEFL='Y') to the last elevated record. We use the retain function to remain the result of NORMEDN for last max TE value to next record. If the next record has different result, the result will be updated, otherwise will be remained until the last elevated records. At the end of the data step, Output the result of

NORMEDN and NORMED of last elevated record.

```
data test2;
    set test1;
    by  subjid paramcd adt atm ;

    length normed $60 _LMTEFL $1;

    if first.paramcd then call missing(_LMTEFL, normedn);
    retain _LMTEFL normedn;

    * #3 check the status of the last max treatment emergert value firstly;
    if LMTEFL = "Y" then _LMTEFL="Y";
    if LMTEFL = "Y" then do;
        if mcrit3mn ge mcrit1mn then do;
            normedn=0;
            normed = "No worsening as compared to the baseline status";
        end;
        else if mcrit3mn lt mcrit1mn and last.paramcd then do;
            normedn=4;
            normed ="No assessment after elevation";
        end;
    end;
```

```
    * #4 then check the status of every record occurred on or after the date of the last max
         treatment emergent record according to the sequence of test start date;
    if _LMTEFL = "Y" then do;
        if mcrit3mn ge mcrit1mn and normedn=0 then do;
            normedn=0;
            normed = "No worsening as compared to the baseline status";
        end;
        else if mcrit3mn ge mcrit1mn and lvotfl eq "Y" then do;
            normedn=1;
            normed ="Returned to baseline status before last IMP dose";
        end;
        else if mcrit3mn ge mcrit1mn and ontrtfl ne "Y"  and normedn eq 1 then do;
            normedn=1;
            normed ="Returned to baseline status before last IMP dose";
        end;
        else if mcrit3mn ge mcrit1mn and ontrtfl ne "Y"  and normedn not in (0,1) then do;
            normedn=2;
            normed ="Returned to baseline status after last IMP dose";
        end;
        else if mcrit3mn lt mcrit1mn and normedn not in ( 0 4 )then do;
            normedn=3;
            normed ="Never returned to baseline status";
        end;
    end;

    * #5 output the final check result;
    if last.paramcd then output;

run;
```

Step 4: Merge the NORMED, NORMEDN to source data by subject ID and parameter code.

```
* #6 merge the final check result with original data;
proc sql;
   create table test3 as select a.*, b.normed,b.normedn
      from test1 as a left join test2 as b
         on a.subjid=b.subjid and a.paramcd=b.paramcd;
quit;
```

Step 5: Set missing value to NORMED, NORMEDN for baseline records since return to baseline status only matter to post baseline records.

```
* #7 set missing value to baseline records since return to baseline status is for
     post baseline records ;
data test3;
   set test3;
   if ablfl='Y' then call missing (normed,normedn);
run;
```

RESULTS

Table 8 displays the dataset generated by above SAS code.

Table 8 NORMED NORMEDN results of dummy data

SUBJID	MCRIT3MN	MCRIT1MN	ADT	ABLFL	ONTRTFL	LMTEFL	LVOTFL	NORMED	NORMEDN
1	1	1	2023-01-01	Y					
1	1	1	2023-01-06		Y	Y		No worsening as compared to the baseline status	0
1	1	0	2023-01-11		Y			No worsening as compared to the baseline status	0
1	1	0	2023-01-17		Y		Y	No worsening as compared to the baseline status	0
2	1	1	2024-01-01	Y					
2	1	2	2024-01-06		Y	Y		Returned to baseline status before last IMP dose	1
2	1	1	2024-01-11		Y		Y	Returned to baseline status before last IMP dose	1
2	1	1	2024-01-17					Returned to baseline status before last IMP dose	1
3	1	1	2024-01-22	Y					
3	1	0	2024-01-27		Y			Returned to baseline status after last IMP dose	2
3	1	1	2024-02-01		Y		Y	Returned to baseline status after last IMP dose	2
3	1	3	2024-02-06			Y		Returned to baseline status after last IMP dose	2
3	1	0	2024-02-11					Returned to baseline status after last IMP dose	2
4	1	1	2023-01-22	Y					
4	1	2	2023-01-27		Y			Never returned to baseline status	3
4	1	2	2023-02-01		Y		Y	Never returned to baseline status	3
4	1	3	2023-02-06			Y		Never returned to baseline status	3
4	1	2	2023-02-11					Never returned to baseline status	3
5	1	1	2023-01-11	Y					
5	1	2	2023-01-17		Y	Y	Y	No assessment after elevation	4

INTRODUCTION TO TABLE OF RETURN TO BASELINE STATUS AND INTERPRETATION OF RESULTS

Based on the variables NORMED and NORMEDN created above, we can generate the table of the subject's ALT return to the baseline status. The data in Table 9 is another set of hypothetical data. The results of ALT serum concentration return to baseline status are shown based on the stratification of the worst ALT concentration (the max value) during the treatment emergent period. N1 stands for the total number of people in each stratification.

As shown in Table 9, there were 10 subjects in each arm whose highest ALT concentration was >3ULN during the treatment emergent period. In arm A, there were 9 subjects' ALT concentration returned to the baseline status, and 1 subject never returned to the baseline status. While in arm B, only 5 subjects returned to the baseline status, and the other 5 subjects never returned. More subjects had worse status of liver than baseline in arm B. This data might suggest that the drug in arm B would have a greater influence on the liver than in arm A.

At the stratification of >5ULN, although there were two subjects in arm A and only one subject in arm B. Both subject in arm A returned to their baseline status, but the one subject in arm B never returned to baseline status. It is not enough to show that arm B is better than arm A in liver safety.

At the stratification of >10ULN, there were one subject in arm B. But the ALT concentration returned to baseline status before the last IMP dose. It is possible that this increase is related to the subject's own health status and does not necessarily mean that it is a drug-induced injury.

At the stratification of >20ULN, the subject in arm A returned to baseline before the last IMP. In arm B, the subject returned to baseline after the last IMP. Compared to drug B, the time of impact from drug to liver is shorter of drug A. It may be a signal that drug A is better in safety.

Table 9 Return to baseline status of the worst elevated Liver function tests

Laboratory Parameter **Worst elevation on the TE period,** **Normalization n/N1 (%)**	**Arm A**	**Arm B**
Alanine Aminotransferase (ALT)		
> 3 ULN	10	10
Returned to baseline status before last IMP dose	5 (50)	2 (20)
Returned to baseline status after last IMP dose	4 (40)	3 (30)
Never returned to baseline status	1 (10)	5 (50)
No assessment after elevation	0	0
> 5 ULN	2	1
Returned to baseline status before last IMP dose	1 (50)	0
Returned to baseline status after last IMP dose	1 (50)	0
Never returned to baseline status	0	1 (100)
No assessment after elevation	0	0
> 10 ULN	0	1
Returned to baseline status before last IMP dose		1 (100)
Returned to baseline status after last IMP dose		0
Never returned to baseline status		0
No assessment after elevation		0
> 20 ULN	1	1
Returned to baseline status before last IMP dose	1 (100)	0
Returned to baseline status after last IMP dose	0	1 (100)
Never returned to baseline status	0	0
No assessment after elevation	0	0

CONCLUSION

This paper aims to track the status of the worst elevated liver function tests result in a dynamic view and

observe whether the tests results return to the baseline status during the drug administration process. During treatment, if the serum concentrations of ALT, AST, ALP, and TBL are several times higher than the upper limit of normal values, and then return to baseline status. This suggests that the drug's reaction to the liver may be temporary and provides more evidence for the drug's safety analysis.

REFERENCES

Guidance for Industry Drug-Induced Liver Injury: Premarketing Clinical Evaluation

https://www.fda.gov/regulatory-information/search-fda-guidance-documents/drug-induced-liver-injury-premarketing-clinical-evaluation

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Yuan Sun
39F, Tower 3, in99 Chengdu Yintai Centre, 1199 North Section of Tianfu Road, High-Tech Zone, Chengdu, P.R.C
Yuan1.Sun@sanofi.com